

Nu Game Engine

Bryan Edds, 2014

Table of Contents

Table of Contents	2
What's It All About?	3
Very Basic	3
Purely-Functional(ish)	3
2d Game Engine	3
F#	3
Getting Started	4
Running the Nu Project (Nu.exe)	5
Basic Nu Start-up Code	6
What is NuEdit?	7
BlazeVector	12
The Game Engine	15
World	15
Screen	16
Transition(s)	16
Group	16
Entity	16
Engine Details	16
Addresses	16
Transformations	16
XDispatchers, XFields, and the Xtension system	17
Facets	17
Assets and the AssetGraph	20
Subsystems and Message Queues	20

What's It All About?

The Nu Game Engine is a **Very Basic, Purely-Functional(ish), 2d Game Engine** written in **F#**.

Let me explain each of those terms –

Very Basic

Nu is very young, and so it has just about no frills. Is there a particle or special effects system? Not yet, I'm afraid. Is there a sophisticated animation system? Again, not yet. However, there is a tile map system that utilizes Tiled#, and there is a physics system that utilizes Farseer Physics. Rendering, audio, and other IO systems are handled in a cross-platform way with SDL2 / SDL2#. In addition to that, there is an asset management system to make sure your game can run on memory-constrained devices such as the iPhone. On top of all that, there is a built-in game editor called NuEdit! So while there are plenty of missing features, you can see they might be worth waiting for, or even building for yourself!

Purely-Functional(ish)

Nu is built on immutable types, and unlike with other game engines, data transformations and state transitions are implemented with copying rather than mutation.

Don't mistake Nu for being slow, however. Notice I said Purely-Functional-ish. The 'ish' means that there are some imperative operations going on in Nu, almost entirely behind the scenes. For example, the Farseer physics system is written in an imperative style in C#, and some parts of Nu are optimized with imperative code as well. Fortunately, nearly all of this will be transparent to you as the user. When writing code that utilizes, feel empowered to write in the pure-functional style.

2d Game Engine

Nu is not a code library. It is a game software framework, and thus sets up a specific way of approaching and thinking about the design of 2d games. Of course, Nu is intended to be a broadly generic toolkit for 2d game development, but there are some design choices that may sometimes constrain you as much as they help you. Figure out how to leverage Nu's design for your game. If it's a complete mismatch, it might be time to consider using something else.

F#

We know what F# is, so why use it? First, and foremost, its cross-platformedness. Theoretically, Nu should run fine on Mono for systems such as Android, iOS, OSX, and *nixes. It definitely runs on .NET for Windows. Note my weasel-word "theoretically"; Nu is still in such an early stage that it has yet to be configured, deployed, or tested on Mono. Nonetheless, since Nu only takes dependencies on cross-platform libraries, there should be no reason why it can't with a little bit of appropriate nudging.

But more on why F#. F# is probably the best mainstream language available for writing a cross-platform functional game engine. Unlike Clojure, F#'s static type system makes the code easier to reason about and dare I say more efficient. Unlike JVM languages, F# allows us to code and debug with Visual Studio. Finally, I speculate

that game developers have more familiarity with the .NET ecosystem than the JVM, so that leverage is right there.

Getting Started

Nu is made available by a GitHub repository located at <https://github.com/bryanedds/FPWorks>. To obtain it, first **fork** the repository's latest **release** to your own GitHub account (register as a new GitHub user if you don't already have an account). Second, **clone** the forked repository to your local machine (instructions here <https://help.github.com/articles/fork-a-repo>). The Nu Game Engine is now yours!

Note: Unlike code libraries that are distributed via NuGet, forking and cloning the FP Works repository at GitHub is how you attain Nu. You will be happy with this once you need to make and debug your own changes to the game engine!


Upon inspecting your clone of the repository, the first thing you might notice about it is that it contains more than just the Nu Game Engine. It also includes the source for the **Aml** programming language, the **Prime** F# code library, the sample game **BlazeVector** (which we'll be studying in this document), and my WIP role-playing game **OmniBlade**. Both Prime and BlazeVector are required to build the BlazeVector solution we'll be opening in this tutorial, and the rest of the stuff is safely ignored.

To open the BlazeVector solution, first make sure to have Visual Studio 2013 installed (or perhaps an earlier version – not tested!) Then navigate to the `./BlazeVector/BlazeVector` folder and open the `BlazeVector.sln` file. Attempt to build the whole solution. If there is a problem with building it, try to figure it out, and failing that, ask me questions via bryanedds@gmail.com.

Running the Nu Project (Nu.exe)

Once you have built the solution, try running the standalone engine by setting the Nu project as the StartUp Project, and then running.

When the app is run from Visual Studio, you'll notice a window popping up that is filled with a nice white color. By default, Nu does nothing but clear the frame buffer with white pixels. There is no interactivity in this program, as the engine is not yet being told to do anything.



Though this is not an interesting program, a look at the code behind it should be enlightening.

Basic Nu Start-up Code

Here's the main code presented with comments -

```
// Nu Game Engine.
// Copyright (C) Bryan Edds, 2013-2014.

namespace Nu
open SDL2
open Nu
open Nu.NuConstants
module Program =

  // this the entry point for the empty Nu application
  let [<EntryPoint>] main _ =

    // this initializes miscellaneous values required by the engine. This should always be the
    // first line in your game program.
    World.init ()

    // this specifies the manner in which the game is viewed. With this configuration, a new
    // window is created with a title of "Nu Game Engine".
    let sdlViewConfig =
      NewWindow
      { WindowTitle = "Nu Game Engine"
        WindowX = SDL.SDL_WINDOWPOS_UNDEFINED
        WindowY = SDL.SDL_WINDOWPOS_UNDEFINED
        WindowFlags = SDL.SDL_WindowFlags.SDL_WINDOW_SHOWN }

    // this specifies the manner in which the game's rendering takes place. With this
    // configuration, rendering is hardware-accelerated and synchronized with the system's
    // vertical re-trace, making for fast and smooth rendering.
    let sdlRendererFlags =
      enum<SDL.SDL_RendererFlags>
      (int SDL.SDL_RendererFlags.SDL_RENDERER_ACCELERATED |||
       int SDL.SDL_RendererFlags.SDL_RENDERER_PRESENTVSYNC)

    // this makes a configuration record with the specifications we set out above.
    let sdlConfig =
      { ViewConfig = sdlViewConfig
        ViewW = ResolutionX
        ViewH = ResolutionY
        RendererFlags = sdlRendererFlags
        AudioChunkSize = AudioBufferSizeDefault }

    // this is a callback that attempts to make 'the world' in a functional programming
    // sense. In a Nu game, the world is represented as a complex record type named World.
    let tryMakeWorld sdlDeps =

      // Game dispatchers specify some unique, high-level behavior and data for your game.
      // Since this particular program has no unique behavior, the vanilla base class
      // GameDispatcher is used.
      let gameDispatcher = GameDispatcher () :> obj

      // here is an attempt to make the world using SDL dependencies that will be created
      // from the invoking function using the SDL configuration that we defined above, the
      // gameDispatcher immediately above, and a value that could have been used to
      // user-defined data to the world had we needed it (we don't, so we pass unit).
      World.tryMakeEmpty sdlDeps gameDispatcher true ()

    // this is a callback that specifies your game's unique behavior when updating the world
    // every tick. The World value is the state of the world after the callback has transformed
    // the one it receives. It is here where we first clearly see Nu's purely-functional(ish)
    // design. The World type is almost entirely immutable, and thus the only way to update it
    // is by making a new copy of an existing instance. Since we need no special update
    // behavior in this program, we simply return the world as it was received.
    let updateWorld world = world
```

```
// after some configuration it is time to run Nu. We're off and running!  
World.run tryMakeWorld updateWorld sdlConfig
```


Hopefully that was somewhat enlightening. You can find this code from Visual Studio in the Program.fs file of the Nu project in the Nu.sln solution.

When creating a new Nu game project, you can copy and modify this file into your project to use as a template for your program. Creating a new Nu game project is mostly just creating a new F# program project, setting up the references, and using said code as a template.

Before discussing Nu's game engine design, let's have a little fun messing around with Nu's real-time interactive game editor, NuEdit.

What is NuEdit?

NuEdit is Nu's fairly usable game editor. Here is a screenshot of an empty editing session –




NOTE: There may still be some stability issues with NuEdit, so save your documents early and often, and for goodness' sake use a source control system!


Run NuEdit by setting the NuEdit project as the StartUp Project in Visual Studio, and then running.

You'll instantly notice an Open File dialog appear from which you are instructed to "Select your game's executable file..." If you select an executable .NET file that contains concrete sub-classes of an Entity2dDispatcher, they will be made available for use in the editor.

First, we'll create a blank button by ensuring that ButtonDispatcher is selected in the combo box to the right of the Create Entity button on the main tool bar, and then pressing the Create Entity button.




You'll notice a squished button appear in the middle of the editing panel. By default, most entities are created with a size of 64x64. Fortunately, Nu gives you an easy way to resize the entity to fit the button's image by pressing the Quick Size button. Press it now.



We have a full button! Notice the property grid on the right got filled with various field names and their corresponding values. These values can be edited manually. For an entity that will be used to control the game's state (like a button), the first thing you will want to do is to give it an appropriate name. Simply double-click the Name field, delete the contents, and then enter the text "MyButton". Naming entities give you the ability to access them at runtime via that name once you have loaded the containing document in your game.

Notice also that you can click and drag on the button to move it about the screen. Once an entity is selected, you can also right-click it for more operations.

Here we've renamed the button and moved it to the bottom right of the screen –




Notice you have the full power of undo and redo. Nonetheless, you should still save your documents often in case this early version of NuEdit goes bananas on you.

Let's now try putting NuEdit in interactive mode so that we can test that our button clicks as we expect. Toggle on the Interact button at the top right, then click on the button.

Once you're satisfied, toggle off the Interact button to return to editing mode.


Now let's make a default tile map to play around with. BUT FIRST, we need to change the depth of our button entity so that it doesn't get covered by the new tile map! Change the value in the button's Depth field to 10.

In the drop down box to the right of the Create Entity button, select (or type) TileMapDispatcher, and then press the Create Entity button, and then click the Quick Size button. You'll get this –



Click and drag the tile map so its bottom-left corner lines up with the top left of the editing panel.

Tile maps, by the way, are created with the free tile map editor Tiled found at <http://www.mapeditor.org/>. All credit to the great chap who made and maintains it!



Now click and drag with the MIDDLE mouse button to change the position of the camera that is used to view the game. Check out your lovely new tile map! If your camera gets lost in space, click the Reset Camera button that is to the left of the Interact button.

Now let's create some blocks to fall down and collide with the tile map using physics. First, we must change the default depth at which new entities are created (again, so the tile map doesn't overlap them). In the at Depth text box to the left of the Quick Size button, type in a 1. In the combo box to the right of the Create Entity button, select (or type) BlockDispatcher, and then click the Create Entity button. You'll see a box created in the middle of the screen that falls directly down.



Notice that you can create blocks in other places by right-clicking at the desired location and then, in the context menu that pops up, clicking Create.



Blocks can be clicked and dragged around like other entities.

We can now save the document for loading into a game by clicking File -> Save...

Lastly, we can add custom fields (known as XFields) to each entity by selecting it on the screen and pressing the Add button in the XFields box atop the property grid. We have no use for this now, however, so we won't click anything further.

Let's watch Nu in action with the sample game called BlazeVector.

BlazeVector

This is the sample game for the Nu Game Engine. In Visual Studio, set the BlazeVector project as the StartUp Project, and then run. We can look at how Nu hangs together a bit by studying BlazeVector's top level code.

First, however, we need to go over the constants that BlazeVector uses. These are defined in the BlazeConstants.fs file –

```
namespace BlazeVector
open Nu
open Nu.NuConstants
module BlazeConstants =

    // misc constants. These, and the following constants, will be explained in depth later. Just
    // scan over them for now, or look at them in the debugger on your own.
    let GuiPackageName = "Gui"
    let StagePackageName = "Stage"
    let StagePlayerName = "Player"
    let StagePlayName = "StagePlay"
    let StagePlayFileName = "Assets/BlazeVector/Groups/StagePlay.nugroup"
    let SectionName = "Section"
    let Section0FileName = "Assets/BlazeVector/Groups/Section0.nugroup"
    let Section1FileName = "Assets/BlazeVector/Groups/Section1.nugroup"
    let Section2FileName = "Assets/BlazeVector/Groups/Section2.nugroup"
    let Section3FileName = "Assets/BlazeVector/Groups/Section3.nugroup"
    let SectionFileNames = [Section0FileName; Section1FileName; Section2FileName; Section3FileName]
    let SectionCount = 32

    // asset constants
    let NuSplashSound = { SoundAssetName = "Nu"; PackageName = GuiPackageName; PackageFileName = AssetGraphFileName }
    let MachinerySong = { SongAssetName = "Machinery"; PackageName = GuiPackageName; PackageFileName = AssetGraphFileName }
    let DeadBlazeSong = { SongAssetName = "DeadBlaze"; PackageName = StagePackageName; PackageFileName = AssetGraphFileName }
    let HitSound = { SoundAssetName = "Hit"; PackageName = StagePackageName; PackageFileName = AssetGraphFileName }
    let ExplosionSound = { SoundAssetName = "Explosion"; PackageName = StagePackageName; PackageFileName = AssetGraphFileName }
    let ShotSound = { SoundAssetName = "Shot"; PackageName = StagePackageName; PackageFileName = AssetGraphFileName }
    let JumpSound = { SoundAssetName = "Jump"; PackageName = StagePackageName; PackageFileName = AssetGraphFileName }
    let DeathSound = { SoundAssetName = "Death"; PackageName = StagePackageName; PackageFileName = AssetGraphFileName }
    let EnemyBulletImage = { ImageAssetName = "EnemyBullet"; PackageName = StagePackageName; PackageFileName = AssetGraphFileName }
    let PlayerBulletImage = { ImageAssetName = "PlayerBullet"; PackageName = StagePackageName; PackageFileName = AssetGraphFileName }
    let EnemyImage = { ImageAssetName = "Enemy"; PackageName = StagePackageName; PackageFileName = AssetGraphFileName }
    let PlayerImage = { ImageAssetName = "Player"; PackageName = StagePackageName; PackageFileName = AssetGraphFileName }

    // transition constants
    let IncomingTimeSplash = 60L
    let IncomingTime = 20L
    let IdlingTime = 60L
    let OutgoingTimeSplash = 40L
    let OutgoingTime = 30L
    let StageOutgoingTime = 90L

    // splash constants
    let SplashAddress = addr "Splash"

    // title constants
    let TitleAddress = addr "Title"
    let TitleGroupFileName = "Assets/BlazeVector/Groups/Title.nugroup"
    let TitleGroupAddress = addr "Title/Group"
    let SelectTitleEventName = addr "Select/Title"
    let ClickTitlePlayEventName = addr "Click/Title/Group/Play"
    let ClickTitleCreditsEventName = addr "Click/Title/Group/Credits"
    let ClickTitleExitEventName = addr "Click/Title/Group/Exit"

    // stage constants
    let StageAddress = addr "Stage"
    let StageGroupFileName = "Assets/BlazeVector/Groups/StageGui.nugroup"
    let StageGroupAddress = addr "Stage/Group"
    let ClickStageBackEventName = addr "Click/Stage/Group/Back"
```

```
// credits constants
let CreditsAddress = addr "Credits"
let CreditsGroupFileName = "Assets/BlazeVector/Groups/Credits.nugroup"
let CreditsGroupAddress = addr "Credits/Group"
let ClickCreditsBackEventName = addr "Click/Credits/Group/Back"
```

Nothing terribly interesting, so let's jump to Program.fs -

```
namespace BlazeVector
open System
open SDL2
open Nu
open Nu.NuConstants
open BlazeVector
module Program =

    // this the entry point for the BlazeVector application
    let [<EntryPoint>] main _ =

        // this initializes miscellaneous values required by the engine. This should always be the
        // first line in your game program.
        World.init ()

        // this specifies the manner in which the game is viewed. With this configuration, a new
        // window is created with a title of "BlazeVector".
        let sdlViewConfig =
            NewWindow
            { WindowTitle = "BlazeVector"
              WindowX = SDL.SDL_WINDOWPOS_UNDEFINED
              WindowY = SDL.SDL_WINDOWPOS_UNDEFINED
              WindowFlags = SDL.SDL_WindowFlags.SDL_WINDOW_SHOWN }

        // this specifies the manner in which the game's rendering takes place. With this
        // configuration, rendering is hardware-accelerated and synchronized with the system's
        // vertical re-trace, making for fast and smooth rendering.
        let sdlRendererFlags =
            enum<SDL.SDL_RendererFlags>
            (int SDL.SDL_RendererFlags.SDL_RENDERER_ACCELERATED |||
             int SDL.SDL_RendererFlags.SDL_RENDERER_PRESENTVSYNC)

        // this makes a configuration record with the specifications we set out above.
        let sdlConfig =
            { ViewConfig = sdlViewConfig
              ViewW = ResolutionX
              ViewH = ResolutionY
              RendererFlags = sdlRendererFlags
              AudioChunkSize = AudioBufferSizeDefault }

        // after some configuration it is time to run the game. We're off and running!
        World.run
            (fun sdlDeps -> BlazeFlow.tryMakeBlazeVectorWorld sdlDeps ())
            (fun world -> world)
            sdlConfig
```

Well, honestly, we've seen most of this before, except the window is titled "BlazeVector" and the world creation callback is `BlazeFlow.tryMakeBlazeVectorWorld` instead of `World.tryMakeEmptyWorld`. Let's investigate into `BlazeFlow.tryMakeBlazeVectorWorld` to learn a little more –

```
namespace BlazeVector
open System
open Prime
open Nu
open Nu.NuConstants
open BlazeVector
open BlazeVector.BlazeConstants
module BlazeFlow =

    // this function handles playing the song "Machinery"
    let handlePlaySongMachinery _ world =
        let world = World.playSong MachinerySong 1.0f 0 world
```

```

(Unhandled, world)

// this function handles playing the stage
let handlePlayStage _ world =
  let world = World.fadeOutSong DefaultTimeToFadeOutSongMs world
  let world = World.transitionScreen StageAddress world
  (Unhandled, world)

// this function adds the BlazeVector title screen to the world
let addTitleScreen world =

  // this adds a dissolve screen from the specified file with the given parameters
  let world = World.addDissolveScreenFromFile typeof<ScreenDispatcher>.Name TitleGroupFileName (List.last TitleGroupAddress) IncomingTime
  OutgoingTime TitleAddress world

  // this subscribes to the event that is raised when the Title screen is selected for
  // display and interaction, and handles the event by playing the song "Machinery"
  let world = World.subscribe4 SelectTitleEventName [] (CustomSub handlePlaySongMachinery) world

  // subscribes to the event that is raised when the Title screen's Play button is
  // clicked, and handles the event by transitioning to the Stage screen
  let world = World.subscribe4 ClickTitlePlayEventName [] (CustomSub handlePlayStage) world

  // subscribes to the event that is raised when the Title screen's Credits button is
  // clicked, and handles the event by transitioning to the Credits screen
  let world = World.subscribe4 ClickTitleCreditsEventName [] (ScreenTransitionSub CreditsAddress) world

  // subscribe4s to the event that is raised when the Title screen's Exit button is clicked,
  // and handles the event by exiting the game
  World.subscribe4 ClickTitleExitEventName [] ExitSub world

  // pretty much the same as above, but for the Credits screen
  let addCreditsScreen world =
    let world = World.addDissolveScreenFromFile typeof<ScreenDispatcher>.Name CreditsGroupFileName (List.last CreditsGroupAddress) IncomingTime
    OutgoingTime CreditsAddress world
    World.subscribe4 ClickCreditsBackEventName [] (ScreenTransitionSub TitleAddress) world

  // and so on.
  let addStageScreen world =
    let world = World.addDissolveScreenFromFile typeof<StageScreenDispatcher>.Name StageGroupFileName (List.last StageGroupAddress) IncomingTime
    StageOutgoingTime StageAddress world
    World.subscribe4 ClickStageBackEventName [] (ScreenTransitionSub TitleAddress) world

  // here we make the BlazeVector world in a callback from the World.run function.
  let tryMakeBlazeVectorWorld sdlDeps extData =

    // our custom game dispatcher here is OmniGameDispatcher
    let gameDispatcher = BlazeVectorDispatcher () :> obj

    // we use World.tryMakeEmpty to create an empty world that we will transform to create the
    // BlazeVector world
    let optWorld = World.tryMakeEmpty sdlDeps gameDispatcher GuiAndPhysicsAndGamePlay extData
    match optWorld with
    | Left _ as left -> left
    | Right world ->

      // hint to the renderer that the Gui package should be loaded up front
      let world = World.hintRenderingPackageUse AssetGraphFileName GuiPackageName world

      // add our UI screens to the world
      let world = addTitleScreen world
      let world = addCreditsScreen world
      let world = addStageScreen world

      // add to the world a splash screen that automatically transitions to the Title screen
      let splashScreenImage = { ImageAssetName = "Image5"; PackageName = DefaultPackageName; PackageFileName = AssetGraphFileName }
      let world = World.addSplashScreenFromData TitleAddress SplashAddress typeof<ScreenDispatcher>.Name IncomingTimeSplash IdlingTime
      OutgoingTimeSplash splashScreenImage world

      // play a neat sound effect, and select the splash screen
      let world = World.playSound NuSplashSound 1.0f world
      let world = World.selectScreen SplashAddress world

      // return our world within the expected Either type, and we're off!
      Right world

```

This gives us a good idea how everything you see in the game is created and hooked together. There are far more details on the game's implementation in `BlazeDispatchers.fs`, but documentation on that is not yet available.

As you might notice in the code shown, there is no mutation going on that is visible to the end-user. Immutability is a cornerstone of Nu's design and implementation. Remember the Undo and Redo features in NuEdit? Those are implemented simply by keeping references to past and future world values, rewinding and


fast-forwarding to them as needed. This approach is a massive improvement over the complicated and fragile imperative ‘Command Design Pattern’ approach.

The Game Engine

You might now have a vague idea of how Nu is used and structured. Let’s try to give you a clearer idea.

First and foremost, Nu was designed for *games*. This may seem an obvious statement, but it has some implications that vary it from other middleware technologies, including most game engines!

Nu comes with an appropriate game structure out of the box, allowing you to house your game’s implementation inside of it. Here’s the overall structure of a game as prescribed by Nu –



In the above diagram, $X \rightarrow Y$ denotes a one-to-many relationship, and $[X] \rightarrow [Y]$ denotes that each X has a one-to-many relationship with Y . So for example, there is only one `Game` in existence, but it can contain many `Screens` (such as a `Title Screen` and a `Credits Screen`). And for each screen, it may contain multiple `Groups`, each under which collections of `Entities` may be cohered.

Everyone should know by now that UIs are an intrinsic part of games. Rather than tacking on a UI system like other engines, Nu implements its UI components directly as entities. There is no arbitrary divide between a `Block` entity in the game and a `Button` entity.

Let’s break down what each of Nu’s most important types mean in detail.

World

We already know a bit about the `World` type. As you can see in the above diagram, it holds the `Game` value. It also holds all the other facilities needed to execute a game such as a rendering context, an audio context, their related message queues (more on this later), a purely-functional message system (far more appropriate to a functional game than .NET’s or even F#’s mutable event systems), and other types of dependencies and configuration values. When you want something in your game to change, you start at the `World` and work your way inward.

Screen

Screens are precisely what they sound like – a way to implement a single ‘screen’ of interaction in your game. In Nu’s conceptual model, a game is nothing more than a series of interactive screens to be traversed like a graph. The main game simulation occurs within a given screen, just like everything else. How screens transition from one to another is specified in code. In fact, we’ve already seen the code that does this in the `BlazeVector.BlazeFlow.addTitleScreen` function that we studied some pages above.

Transition(s)

Screen transitions in other engines, like their UIs, are typically tacked on - if present at all. However, Nu knows that no game wants to move from one screen to another without some sort of pleasant graphical transition sequence, so the concept is built right in the engine.

There are two Transitions per screen; one describes how the screen transitions in from other screens, and the other describes how it transitions out to other screens.

Group

A Group represents a logical ‘grouping’ of entities. NuEdit actually builds one group of entities at a time. At run-time, multiple of those groups can have their files loaded into a single screen.

Entity

And here we come down to brass tacks. Entities represent individual interactive things in your game. We’ve seen several already – a button, a tile map, and blocks. What differentiates a button entity from a block entity, though? Each entity picks up its unique attributes from its `XDispatcher`. What is a `XDispatcher`? Well, it’s a little complicated, so we’ll touch on that later!

Engine Details

Addresses

You may be wondering about the details of connecting code-driven behavior to entities created in the editor and loaded from a file at run-time. Accessing entities, including the ones loaded from a file is done with Nu’s realization of ‘addresses’. Each entity has an address of the form ‘ScreenName/GroupName/EntityName’, where the ScreenName is the name that is given to the containing Screen value, GroupName is the name given to the containing Group value, and Entity name is the name given to the Entity. Remember how we changed the Name field of the button object that we created to “MyButton” earlier in this document? That’s what we’re talking about, and the entity’s name is just the last part of its address.

Transformations

Given all this, how do we actually make transformations to a given entity in the world?

Well, first we need to find the thing in the world that we want to transform. Then we have to transform it, and then finally place the transformed value unto a new copy of the world.

Here's some code that grabs an entity at a specific address using the `getEntity` function –

```
let buttonAddress = addr "TitleScreen/MainGroup/MyButton"
let button = World.getEntity buttonAddress world
```

Note that in this (and in the following code) we presume that both the Prime and OpenTK namespaces are open.

This will return an entity value at the given address. Now let's transform that button, say, by disabling it.

```
let button = { button with Enabled = false }
```

That works for a simple field that exists on all entities. However, Nu uses a system where fields that are unique to different entities are dynamically created and accessed. Since we know we have a button, we know that it has a `Position` field that can be set. Let's write that code now –

```
let button = Entity.setPosition (Vector2 (100.0f, 100.0f)) button
```

Finally, we place the transformed value unto a new copy of the old world using the `setEntity` function –

```
let world = World.setEntity buttonAddress button world
```

XDispatchers, XFields, and the Xtension system

TODO.

Facets

Due to the special capabilities provided by Xtensions, users can use functions to compose new XDispatchers to implement complex Entity, Group, or Screen behaviors. In Nu, a module whose functions define a new behavior is called a 'Facet'. Complex behavior for a single Entity, Group, or Screen can be defined by composing multiple facets in an XDispatcher definition. Let's take a look at the definition and use of a simple facet now –

```
[<AutoOpen>]
module SimpleSpriteFacetModule =

    type Entity with

        member entity.SpriteImage with get () = entity? SpriteImage () : Image
        static member setSpriteImage (value : Image) (entity : Entity) : Entity = entity? SpriteImage <- value

[<RequireQualifiedAccess>]
module SimpleSpriteFacet =

    let init (entity : Entity) (_ : IXDispatcherContainer) =
        Entity.setImage { ImageAssetName = "Image3"; PackageName = DefaultPackageName; PackageFileName = AssetGraphFileName } entity

    let getRenderDescriptors entity viewType world =
        if entity.Visible && Camera.inView3 entity.Position entity.Size world.Camera then
```

```

[LayerableDescriptor
{ Depth = entity.Depth
  LayeredDescriptor =
    SpriteDescriptor
    { Position = entity.Position
      Size = entity.Size
      Rotation = entity.Rotation
      ViewType = viewType
      OptInset = None
      Image = entity.Image
      Color = Vector4.One }}]

else []

let getQuickSize (entity : Entity) world =
let image = entity.Image
match Metadata.tryGetTextureSizeAsVector2 image.ImageAssetName image.PackageName world.AssetMetadataMap with
| None -> DefaultEntitySize
| Some size -> size

```

This SimpleSpriteFacet is used to define simple sprite rendering behavior for an Entity, and may be used to define a new dispatcher like so –

```

[<AutoOpen>]
module SimpleSpriteDispatcherModule =

type [<Sealed>] SimpleSpriteDispatcher () =
inherit Entity2dDispatcher ()

override dispatcher.Init (entity, dispatcherContainer) =
let entity = base.Init (entity, dispatcherContainer)
SimpleSpriteFacet.init entity dispatcherContainer

override dispatcher.GetRenderDescriptors (entity, world) =
SimpleSpriteFacet.getRenderDescriptors entity Relative world

override dispatcher.GetQuickSize (entity, world) =
SimpleSpriteFacet.getQuickSize entity world

```

Similar to the SimpleSpriteFacet, there is also a SimpleBodyFacet. However, instead of defining a sprite-displaying behavior, the SimpleBodyFacet defines simple physics behavior for an entity.

```

[<AutoOpen>]
module SimpleBodyFacetModule =

type Entity with

member entity.MinorId with get () = entity?MinorId () : Guid
static member setMinorId (value : Guid) (entity : Entity) : Entity = entity?MinorId <- value
member entity.BodyType with get () = entity?BodyType () : BodyType
static member setBodyType (value : BodyType) (entity : Entity) : Entity = entity?BodyType <- value
member entity.Density with get () = entity?Density () : single
static member setDensity (value : single) (entity : Entity) : Entity = entity?Density <- value
member entity.Friction with get () = entity?Friction () : single
static member setFriction (value : single) (entity : Entity) : Entity = entity?Friction <- value
member entity.Restitution with get () = entity?Restitution () : single
static member setRestitution (value : single) (entity : Entity) : Entity = entity?Restitution <- value
member entity.FixedRotation with get () = entity?FixedRotation () : bool
static member setFixedRotation (value : bool) (entity : Entity) : Entity = entity?FixedRotation <- value
member entity.LinearDamping with get () = entity?LinearDamping () : single
static member setLinearDamping (value : single) (entity : Entity) : Entity = entity?LinearDamping <- value
member entity.AngularDamping with get () = entity?AngularDamping () : single
static member setAngularDamping (value : single) (entity : Entity) : Entity = entity?AngularDamping <- value
member entity.GravityScale with get () = entity?GravityScale () : single
static member setGravityScale (value : single) (entity : Entity) : Entity = entity?GravityScale <- value
member entity.CollisionCategories with get () = entity?CollisionCategories () : string
static member setCollisionCategories (value : string) (entity : Entity) : Entity = entity?CollisionCategories <- value
member entity.CollisionMask with get () = entity?CollisionMask () : string
static member setCollisionMask (value : string) (entity : Entity) : Entity = entity?CollisionMask <- value
member entity.IsBullet with get () = entity?IsBullet () : bool
static member setIsBullet (value : bool) (entity : Entity) : Entity = entity?IsBullet <- value
member entity.IsSensor with get () = entity?IsSensor () : bool
static member setIsSensor (value : bool) (entity : Entity) : Entity = entity?IsSensor <- value

static member getPhysicsId (entity : Entity) = PhysicsId (entity.Id, entity.MinorId)

```

```
[<RequireQualifiedAccess>]
module SimpleBodyFacet =

    let init (entity : Entity) (_ : IXDispatcherContainer) =
        entity |>
            Entity.setMinorId (NuCore.makeId ()) |>
            Entity.setBodyType BodyType.Dynamic |>
            Entity.setDensity NormalDensity |>
            Entity.setFriction 0.0f |>
            Entity.setRestitution 0.0f |>
            Entity.setFixedRotation false |>
            Entity.setLinearDamping 1.0f |>
            Entity.setAngularDamping 1.0f |>
            Entity.setGravityScale 1.0f |>
            Entity.setCollisionCategories "1" |>
            Entity.setCollisionMask "*" |>
            Entity.setIsBullet false |>
            Entity.setIsSensor false

    let registerPhysics getBodyShape address world =
        let entity = World.getEntity address world
        let bodyProperties =
            { Shape = getBodyShape entity world
              BodyType = entity.BodyType
              Density = entity.Density
              Friction = entity.Friction
              Restitution = entity.Restitution
              FixedRotation = entity.FixedRotation
              LinearDamping = entity.LinearDamping
              AngularDamping = entity.AngularDamping
              GravityScale = entity.GravityScale
              CollisionCategories = Physics.toCollisionCategories entity.CollisionCategories
              CollisionMask = Physics.toCollisionCategories entity.CollisionMask
              IsBullet = entity.IsBullet
              IsSensor = entity.IsSensor }
        let physicsId = Entity.getPhysicsId entity
        let position = entity.Position + entity.Size * 0.5f
        let rotation = entity.Rotation
        World.createBody address physicsId position rotation bodyProperties world

    let unregisterPhysics address world =
        let entity = World.getEntity address world
        World.destroyBody (Entity.getPhysicsId entity) world

    let propagatePhysics getBodyShape address world =
        let world = unregisterPhysics address world
        registerPhysics getBodyShape address world

    let handleBodyTransformMessage address (message : BodyTransformMessage) world =
        let entity = World.getEntity address world
        let entity =
            entity |>
                Entity.setPosition (message.Position - entity.Size * 0.5f) |> // TODO: see if this center-offsetting can be
                encapsulated within the Physics module!
                Entity.setRotation message.Rotation
        World.setEntity message.EntityAddress entity world
```

Here's an entity dispatcher that combines the SimpleSpriteFacet and SimpleBodyFacet behaviors –

```
[<AutoOpen>]
module SimpleBodySpriteDispatcherModule =

    type [<Sealed>] SimpleBodySpriteDispatcher () =
        inherit Entity2dDispatcher ()

        let getBodyShape (entity : Entity) _ =
            BoxShape { Extent = entity.Size * 0.5f; Center = Vector2.Zero }

        override dispatcher.Init (entity, dispatcherContainer) =
            let entity = base.Init (entity, dispatcherContainer)
            let entity = SimpleSpriteFacet.init entity dispatcherContainer
            SimpleBodyFacet.init entity dispatcherContainer

        override dispatcher.Register (address, world) =
```

```

SimpleBodyFacet.registerPhysics getBodyShape address world

override dispatcher.Unregister (address, world) =
    SimpleBodyFacet.unregisterPhysics address world

override dispatcher.PropagatePhysics (address, world) =
    SimpleBodyFacet.propagatePhysics getBodyShape address world

override dispatcher.HandleBodyTransformMessage (address, message, world) =
    SimpleBodyFacet.handleBodyTransformMessage address message world

override dispatcher.GetRenderDescriptors (entity, world) =
    SimpleSpriteFacet.getRenderDescriptors entity Relative world

override dispatcher.GetQuickSize (entity, world) =
    SimpleSpriteFacet.getQuickSize entity world

```

By creating new dispatchers by composing facets, arbitrarily complex Entities, Groups, and Screens can be created with relative ease.

Assets and the AssetGraph

TODO.

Serialization and Overlays

TODO.

Subsystems and Message Queues

TODO.